SolidML で簡単にモデル生成

![2018-12-15_22h06_08.png](https://qiita-image-store.s3.amazonaws.com/0/2585/aecc4e17-8e6a-fde2-652f-682c2fa2116c.png)
[SolidML Demonstration](https://keim.github.io/SolidML/demo/index.html?s=%40bg%5B%23679%2C%23fff%2C%23eef%5D%0D%0A%40400%0D%0A3%7Brz120%7DR1%0D%0A3%7Brz120%7DR2%0D%0A%23R1%7B%0D%0A%20%20%7Bx1.3rx1.57rz6ry3s.99h.4sat.99%7DR1%0D%0A%20%20%7Bs4%7Dsphere%0D%0A%7D%0D%0A%23R2%20%7B%0D%0A%20%20%7Bx-1.3rz6ry3s.99h.4sat.99%7DR2%0D%0A%20%20%7Bs4%7Dsphere%0D%0A%7D%0D%0A)

#はじめに
[WebGL Advent Calendar 2018](https://qiita.com/advent-calendar/2018/webgl) の16日目の記事です。


[keim](https://twitter.com/mike_at_cg)と申します。一年くらい前までは [shadertoy](https://www.shadertoy.com/user/keim) や [GLSLfan](https://glslfan.com/?channel=-LNQU8tRFDZxxYe9vo37) など Shader Art 界隈で遊んでおり、去年のアドベントカレンダーでは[「レイマーチングでガラスレンダリング」](https://qiita.com/keim_at_si/items/2fb0ff70527afdd8fd2e)の記事を書いたり、今年は縁あって [GLSL Tech Night 2018](https://connpass.com/event/97565/) で [スピーカーとして登壇](https://www.slideshare.net/keimesuda/glsltech2018)させていただいたりしました。
ここ数か月は主に [three.js](https://threejs.org/) で遊んでおり、ちょうど最近作っていたものが出来上がりつつあったので、この場を借りて公開させていただこうと思います。

#Structure Synth
2007年ころ登場した[Structure Synth](http://structuresynth.sourceforge.net/)というアプリをご存知でしょうか？


複雑な幾何学立体を非常に簡潔なスクリプトで生成することができるソフトです。[Frickrグループ](https://www.flickr.com/groups/structuresynth/)には Structure Synth で生成した立体のレンダリング結果が大量に置かれており、どんな事ができるか何となく解ると思います。
開発者は、フラクタル界隈ではかなり有名な[Syntopia blog](http://blog.hvidtfeldts.net/index.php)というブログを運営されている[Mikael H Christensen](https://twitter.com/SyntopiaDK)氏で、2010年まで開発され、現在は更新を停止しています。
2011年からは Fragment Shader によるレイトレースを利用した[Fragmentarium](http://syntopia.github.io/Fragmentarium/)の開発に移行しており、こちらはポリゴンの制約を受けない本格的なフラクタル立体を描く事ができます。


GPUの性能が向上し Shader レイトレースのノウハウもだいぶ蓄積されてきた現在、フラクタル形状を生成するのにポリゴンベースのStructure Synth はあまり良い選択肢とは言えません。しかし、ポリゴンは汎用的な技術のため一般的な3D描写との相性が良く、オブジェクト数が制限されたフラクタルの独特の形状は他に類のない魅力があり、いまだにコアなファンの支持を受けています。

#SolidML
この Structure Synth のスクリプトである [Eisen Script](http://structuresynth.sourceforge.net/reference.php) を使って three.js の BufferGeometry を生成してみたいと思い、今年10月くらいから実装を開始しました。
11月下旬には一通りの実装を終えたのですが、得意な[MML](https://ja.wikipedia.org/wiki/Music_Macro_Language)と似ていると感じ、自分が書きやすいようにシンタックスシュガーを適当に入れていくうちに別言語の様相を呈してきました。
そこで、それまで StructureSynth.js という名前だったものを SolidML と改名してリリースすることにしました。

ここでは SolidML の簡単な利用方法を紹介します。まだ作り途中のため、ドキュメント整備まで手が回っていませんが、徐々に準備していきたいと考えています。


##環境
ブラウザで https://keim.github.io/SolidML/demo/index.html にアクセスして左のテキストエリアにスクリプトを打って Build ボタンを押せば生成することができます。ビルドに成功すると url query にスクリプトの平文が含まれるので、[このように](https://keim.github.io/SolidML/demo/index.html?s=%40ao%5B1%2C0.7%5D%0A%40mat%5B80%2C20%2C50%2C20%5D%0A%7B%236d6%7D%0A9%7Ba0.95rx30s1.2%7DS%0A%23S%7B36%7Bs0.9x2ry10%7DR%7D%0A%23R%7B19%7Bx0.5rz20rx20%7Dmesh%7D)リンクでスクリプトをシェアすることができます。

##API
https://keim.github.io/SolidML/jsdoc/index.html に一通り書いてありますが、
- [Github レポジトリ](https://github.com/keim/SolidML) の [src/SolidML.js](https://github.com/keim/SolidML/blob/master/src/SolidML.js) と [src/SolidMLBufferGeometry.js](https://github.com/keim/SolidML/blob/master/src/SolidMLBufferGeometry.js) を呼び込む。
- [SolidML.BufferGeometry](https://keim.github.io/SolidML/jsdoc/SolidML.BufferGeometry.html) のコンストラクタに Eisen Script 文字列を渡して、インスタンスを作成する。
これでthree.js 標準の BufferGeometry と同じ感覚で利用することができます。

```javascript:solidml.js
// コンストラクタに Solid ML Script を渡す
const geometry = new SolidML.BufferGeometry( "20{rx18y2s0.90}R#R@37{tube:6{x1rz10}R}" );
// 頂点カラーを有効にする
const material = new THREE.MeshPhysicalMaterial( { vertexColors: THREE.VertexColors } );
// Mesh 生成
const mesh = new Mesh( geometry, material );
```

##構文
基本的には Eisen Script にシンタックスシュガーや追加コマンドを加えたものなので、Eisen Script はそのままでも動きます（[Structure Synth の解説ページ](http://structuresynth.sourceforge.net/learn.php)）。ここでは SolidML の記法をベースに解説したいと思います。
（本家との比較は、[README.md](https://github.com/keim/SolidML/blob/master/README.md)を参照ください。）

### 1.プリミティブを生成する
```eisenscript:box.es
box
```
[![solidml.jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/8c3b6341-9100-a971-986c-f61b6b4c0a3c.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=box%0D%0A)


`box`、`shpere`、`line`、`grid`、`cylinder`、`cone`、`torus`、`tetra`、`octa`、`dodeca`、`icosa`、`triangle`、`mesh`、`tube`、`cmesh`、`ctube` の予約語を書くとそれぞれに対応した大きさ１の形状を置くことができます。
`line`、`cylinder`、`cone`、`torus` など軸が存在する形状はx軸が基準となります。
`shpere`、`line`、`cylinder`、`cone`は、`shpere:20`とコロンの後に正数を指定する事でポリゴン分割数を指定することができます。`line`、`grid`はコロン後の数字で太さのパーセンテージを指定できます。`grid:2`で全体の2%の太さのグリッド形状を置きます。また、`torus[20,8,32]`と記述すると「直径の20%の太さで、断面方向分割数=8、経分割数=32」のトーラス形状を置きます。
[![solidml (1).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/c3b517f2-286b-674a-b553-d6408a379653.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=box%0D%0A%7Bx2%7Dsphere%0D%0A%7Bx4%7Dline%0D%0A%7Bx6%7Dgrid%0D%0A%7Bx8%7Dcylinder%0D%0A%7Bx10%7Dcone%0D%0A%7Bx12%7Dtorus%0D%0A%7Bx14%7Dtetra%0D%0A%7Bx16%7Docta%0D%0A%7Bx18%7Ddodeca%0D%0A%7Bx20%7Dicosa%0D%0A)

### 2.変換＆複製する
```eisenscript:translate.es
10{x2}sphere
```
[![solidml (1).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/37c6acf4-4b8e-1c67-3353-f95ff085afb1.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=10%7Bx2%7Dsphere%3A20%0D%0A)

予約語やルール名の前に `{...}`で変換方法を指定し、その前に変換回数を指定することで、指定回数分変換を繰り返しつつ、形状を置きます。上記スクリプトでは「10回 xを+2移動させつつ 球を配置する」ので、以下のような形状を生成します。

変換コマンドには、軸並行移動（`x`、`y`、`z`）、軸回転（`rx`、`ry`、`rz`）、拡大縮小（`s`）、3x3行列指定（`m`）、軸反転（`fx`、`fy`、`fz`）があります。[本家コマンド](http://structuresynth.sourceforge.net/reference.php#transformations)と同じです。連続した変換は中かっこ内に逆順（行列をかける方向）で書きます。例えば「y軸中心に10度回転してx方向に2移動」は `{x2ry10}` と記述します。

```eisenscript:transandrot.es
10{x2ry10}box
```
[![solidml.jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/ceaec700-1966-fc45-ce62-50a313e3b0a0.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%20%20%20%2010%7Bx2ry10%7Dbox%0D%0A)

変換を分けて連続で記述すると全ての組み合わせを実行します。以下の例では「x+2を5回」と「y+2を4回」で全変換をかけるため、全部で20個の`cylinder`が並びます。

```eisenscript:transandrot.es
5{x2}4{y2}cylinder
```
[![solidml (2).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/6f0f84de-799d-87d1-0f57-af1b199707df.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%20%20%20%205%7Bx2%7D4%7By2%7Dcylinder)


### 3.色操作
変換コマンドと同様に、色変換または色指定を記述することができます。本家と同様に[SVG keyword names](https://www.december.com/html/spec/colorsvg.html)、cssと同様の16進数カラー表現、HSBによるカラー表、おぼびalpha指定が記述可能です。

```eisenscript:colortransform.es
{x0 red}  box
{x2 navy}  sphere
{x4 #ff00ff} grid
{x6 #0f0} cylinder
{x8 h60sat1b1} cone
{x10 a0.5} torus
```
[![solidml (9).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/8076ecae-cddd-23fd-1404-ffddbc36631a.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%7Bx0%20red%7Dbox%0D%0A%7Bx2%20navy%7Dsphere%0D%0A%7Bx4%20%23ff00ff%7Dgrid%0D%0A%7Bx6%20%230f0%7Dcylinder%0D%0A%7Bx8%20h60%7Dcone%0D%0A%7Bx10%20a0.5%7Dtorus)

`h`、`sat`、`b`、`a` の各コマンドはひとつ前のオブジェクト設置時の色からの変化分を記述します。色相は加算、それ以外は乗算されます。初期値は{hue:0, saturation:1, brightness:1}の赤です。
`*`コマンドは色のブレンディングを計算します。例えば`*blue 0.1`であれば、一個オブジェクトを設置するたびに0.1だけ青成分を混ぜます（[本家blendコマンド](http://structuresynth.sourceforge.net/reference.php#transformations)のシンタックスシュガーです）。
また、彩度・明度のみのブレンドも可能です。`*s0.8 0.1`は、彩度0.8に向けて0.1だけ近づけます（本家では彩度明度は乗算しかできないため、目的値に向かって値を近づける方法が提供されていませんでした。）

```eisenscript:colortransform.es
{y0} 36{x1 h10} sphere
{y2} 36{x1 sat0.9} tetra
{y4} 36{x1 b0.9} box
{y6} 36{x1 a0.9} octa
{y8} 36{x1 *blue 0.1} dodeca
{y10sat0.01} 36{x1 *s0.8 0.1} icosa
```
[![solidml (8).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/736de1d4-6c54-6557-4346-7a4faa9de51e.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%7By0%7D%2036%7Bx1h10%7D%20sphere%0D%0A%7By2%7D%2036%7Bx1sat0.9%7D%20tetra%0D%0A%7By4%7D%2036%7Bx1b0.9%7D%20box%0D%0A%7By6%7D%2036%7Bx1a0.9%7D%20octa%0D%0A%7By8%7D%2036%7Bx1%20*blue%200.1%7D%20dodeca%0D%0A%7By10sat0.01%7D36%7Bx1%20*s0.8%200.1%7D%20icosa)

`#?`コマンドを用いると、あらかじめ`@cp[...]`（color pool コマンド）で指定した色のリストからランダムに色を採用します（本家`color random`と`set color pool list:`のシンタックスシュガーです。）color pool は色リストのほかに `@cp:h0`(色相のみ乱数）、`@cp:g0`（グレイスケールの中から乱数）があります。無指定の場合、RGB3値とも乱数で決定します。本家のcolor poolと同等の機能実装で、仕様は[開発者ブログ](http://blog.hvidtfeldts.net/index.php/2009/04/random-colors-color-pools-and-dual-mersenne-twister-goodness/)に説明があります。

```eisenscript:colorpool.es
@cp[black,yellow,yellow]
10{x2#?} cone
```
[![solidml (10).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/97a01f29-75f5-4d62-ee4f-7dd84d68b1ba.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%40cp%5Bblack%2Cyellow%2Cyellow%5D%0A10%7Bx2%23%3F%7D%20cone)


### 4.メッシュ、チューブ
`mesh`、`tube`、`cmesh`、`ctube`の各コマンドは、連続で呼び出されると変換の進む方向に向かってメッシュを作ります。`mesh`はスケーリングが線の細さに影響しますが、`tube`は一定の太さでメッシュを生成します。また、`mesh`はフラットシェーディングによって角が現れるのに対して`cmesh`はスムースシェーディングのため側面が円筒状に見えます（断面は多角形です）。
これらのコマンドは、`shpere`と同様、コロンで区切って数字を指定することで、断面の正多角形数を指定することができます。連続しない別メッシュを作りたい場合は、ルールを分けて記述します。

```eisenscript:mesh.es
100{x1ry10s0.99} mesh
```
[![solidml (11).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/1942470e-bd86-885d-8ced-4565fee81bb4.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=100%7Bx1ry10s0.99%7D%20mesh%0A)


### 5.ルール
`#[rule名]{...}` と記述することで、ひとかたまりの形状配置に対して名前を付けて、１つのプリミティブのように呼び出すことができます。基本的に[本家のルール記述](http://structuresynth.sourceforge.net/reference.php)のシンタックスシュガーです。

```eisenscript:rule.es
{y4 red} BoxAndSphere
{y2 green} BoxAndSphere
{blue} BoxAndSphere

#BoxAndSphere{
  5{x3}box
  {x1.5}4{x3}sphere
}
```
[![solidml (7).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/4fb3dae9-2fe6-2ff7-ec4c-cef2137d5aa7.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%7By4%20red%7D%20BoxAndSphere%0D%0A%7By2%20green%7D%20BoxAndSphere%0D%0A%7Bblue%7D%20BoxAndSphere%0D%0A%0D%0A%23BoxAndSphere%7B%0D%0A%20%205%7Bx3%7Dbox%0D%0A%20%20%7Bx1.5%7D4%7Bx3%7Dsphere%0D%0A%7D)

同名のルールを定義すると、参照されるたびにランダムに選択されます。3つ以上でも同様です。

```eisenscript:rule2.es
{y4 red} 9{x1.5} BoxOrSphere
{y2 green} 9{x1.5} BoxOrSphere
{blue} 9{x1.5} BoxOrSphere

#BoxOrSphere{
  box
}
#BoxOrSphere{
  sphere
}
```
[![solidml (8).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/78b6a28c-2d5e-8bd5-c66e-a661e89dfb6c.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%7By4%20red%7D%209%7Bx1.5%7D%20BoxOrSphere%0D%0A%7By2%20green%7D%209%7Bx1.5%7D%20BoxOrSphere%0D%0A%7Bblue%7D%209%7Bx1.5%7D%20BoxOrSphere%0D%0A%0D%0A%23BoxOrSphere%7B%0D%0A%20%20box%0D%0A%7D%0D%0A%23BoxOrSphere%7B%0D%0A%20%20sphere%0D%0A%7D%0D%0A)

上記ランダム選択時に、ルール名の後に`@w[正数]`を指定することでランダム選択に重み付けをすることができます。この例では、90%の確率で上、10%の確率で下のルールが選択されます。

```eisenscript:rule3.es
9{x1.5}9{y1.5} BoxOrSpheres

#BoxOrSpheres@w9{
  box
}
#BoxOrSpheres@w1{
  8{z0.6}sphere
}
```
[![solidml (10).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/d88bcb10-f56c-d5c8-f631-5e0bac37375b.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=9%7Bx1.5%7D9%7By1.5%7D%20BoxOrSpheres%0D%0A%0D%0A%23BoxOrSpheres%40w9%7B%0D%0A%20%20box%0D%0A%7D%0D%0A%23BoxOrSpheres%40w1%7B%0D%0A%20%208%7Bz0.6%7Dsphere%0D%0A%7D%0D%0A)

ランダム選択されるルールを部分的に差し替えたい場合、`(...|...)`と記述すると複数ルールを一度に定義できます。

```eisenscript:rule4.es
@cp[red,blue,green,yellow]

9{x1.5}9{y1.5} shapes

#shapes{
  {#?} (box|sphere|tetra|octa)
}
```
[![solidml (12).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/4ac6a9f9-e1f6-de72-7f04-7417ba16114f.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=%40cp%5Bred%2Cblue%2Cgreen%2Cyellow%5D%0D%0A%0D%0A9%7Bx1.5%7D9%7By1.5%7D%20shapes%0D%0A%0D%0A%23shapes%7B%0D%0A%20%20%7B%23%3F%7D%20(box%7Csphere%7Ctetra%7Cocta)%0D%0A%7D%0D%0A)

これもシンタックスシュガーの一種で、内部的にはコンパイル前に文字列処理によって以下のように展開されています（この展開結果は、コンソールにログとして出力されますので、デベロッパーツールで確認することができます。）。

```eisenscript:rule4expand.es
@cp[red,blue,green,yellow]

9{x1.5}9{y1.5} shapes

#shapes  @w1{
  {#?}  box 
}
#shapes  @w1{
  {#?}  sphere 
}
#shapes  @w1{
  {#?}  tetra 
}
#shapes  @w1{
  {#?}  octa 
}
```

複数の分岐を記述した場合、全組み合わせが展開されます。沢山の分岐を含めると指数的にルール定義が増えてしまうので注意してください。それぞれの分岐に重み付けを行いたい場合は、分岐内に`@w`コマンドを含めます。

```eisenscript:rule5.es
9{x1.5}9{y1.5} shapes

#shapes{
  {white} (@w4box|@w3icosa)
  {z1.2 s0.5} (@w9{black}sphere|{red}box)
}
```
内部的には、以下のように展開されます。

```eisenscript:rule5expand.es
9{x1.5}9{y1.5} shapes

#shapes    @w36{
  {white}  box 
  {z1.2 s0.5}  {black}sphere 
}
#shapes    @w4{
  {white}  box 
  {z1.2 s0.5}  {red}box 
}
#shapes    @w27{
  {white}  icosa 
  {z1.2 s0.5}  {black}sphere 
}
#shapes    @w3{
  {white}  icosa 
  {z1.2 s0.5}  {red}box 
}
```
[![solidml (15).jpg](https://qiita-image-store.s3.amazonaws.com/0/2585/4bd59442-6ffd-d663-cb3a-b5e5ddfb02ed.jpeg)
script](https://keim.github.io/SolidML/demo/index.html?s=9%7Bx1.5%7D9%7By1.5%7D%20shapes%0D%0A%0D%0A%23shapes%7B%0D%0A%20%20%7Bwhite%7D%20(%40w4box%7C%40w3icosa)%0D%0A%20%20%7Bz1.2%20s0.5%7D%20(%40w9%7Bblack%7Dsphere%7C%7Bred%7Dbox)%0D%0A%7D)


### 6.ルールの再帰呼び出し

（To Be Added）


### 6.システムコマンド
`@1000` max depth。再帰呼び出しの最大回数
`@mo10000` max object。オブジェクトの最大数
`@s1234` 乱数のシード値の指定。
`@bg[#679,#fff,#eef]` 背景の空、床、チェッカーボードの各色を指定。
`@mat[10,90,30,20]` 素材の metalness, roughness, clearcoat, clearcoat roughness を％で指定。

（To Be Added）


# まとめ
（すいません。後半間に合わず、中途半端な解説でリリースしてしまいました。後日追記します。）
SolidML は、Struture Synth の Three.js への拡張実装です。URLをもちいて比較的簡単にシェアできますので、面白い形状ができたらSNSで拡散したり、Gist で書き捨てしたりしてくれると嬉しいです。
この技術はうまく使えば、例えばメカパーツをベースに生成することでシューティングゲームのボス生成とか、応用範囲が結構広いのではないかと考えています。もうちょっと色々展開できたらと思っています。